Inhomogeneous Linear Recurrence Equations: Method of Undetermined Coefficients

Date: Nov 13, 2001
Last Revision: Nov 15, 2001

Bent E. Petersen
bent@alum.mit.edu
petersen@math.orst.edu

Course: Mth 355 (a.k.a. Mth 399)
Term: Fall 2001
File name: `355f2001_inhrec.mws`
Assignment 7 (problems at end) - due Nov 21, 2001

```
> restart;
```

Introduction

Maple has no difficulty solving simple inhomogeneous linear recurrence equations. For example,

```
> eqn0:=a(n)=4*a(n-1)-4*a(n-2)+3^n+4^n;
```

$$eqn0 := a(n) = 4\,a(n-1) - 4\,a(n-2) + 3^n + 4^n$$

```
> init0:=a(0)=A,a(1)=B;
```

$$init0 := a(0) = A,\ a(1) = B$$

```
> soln0:=rsolve({eqn0,init0},a(n));
```

$$soln0 := \left(-A + \frac{1}{2}B\right)(n+1)\,2^n - \left(\frac{1}{2}B - 2A\right)2^n + \left(-\frac{17}{2}n - \frac{17}{2}\right)2^n - \frac{9}{2}2^n + 9\,3^n + 4\,4^n$$

This means

```
> a(n)=soln0;
```

$$a(n) = \left(-A + \frac{1}{2}B\right)(n+1)\,2^n - \left(\frac{1}{2}B - 2A\right)2^n + \left(-\frac{17}{2}n - \frac{17}{2}\right)2^n - \frac{9}{2}2^n + 9\,3^n + 4\,4^n$$

Very convenient, but not very illuminating! After all, of what importance is the solution to a contrived problem? It is the underlying ideas and the method of solution that are important.

If we were doing this example by hand we would first find the characteristic polynomial of the associated homogeneous equation. Then we would observe the characteristic roots are 2, 2. In particular, 3 and 4 are not characteristic. Thus we would expect a particular solution of the form A 3^n + B 4^n. We would substitute this trial solution and find A = 9 and B = 4 (as we see above). This is not so bad, but it can involve a lot of algebra.

In this worksheet we get Maple to do the algebra. Nothing we do here is needed if we just want a solution! The purpose of this worksheet is to provide support for hand calculation so we can see how things work without being distracted by algebra errors.

It will be convenient, but not necessary, to have a procedure to normalize a polynomial, that is, to divide by the leading coefficient to make the polynomial monic. The degree() function fails easily so monic() is a bit tricky to get to work. In addition we have to simplify the result to avoid problems with roots() later.

```
> monic:=proc(p,z)
>   local k,c,q;
>   q:=sort(collect(p,z),z);
>   k:=degree(simplify(q),z);
>   c:=coeff(q,z,k);
>   return(simplify(q/c));
> end:
```

Method of Undetermined Coefficients

The procedures presented here are not very robust. In particular we assume our linear recurrence equations are in normal form.

```
> `a(n) = c[1]*a(n-1) + ... +c[m]*a(n-m) + f(n)`;
```

$$a(n) = c[1]*a(n-1) + ... +c[m]*a(n-m) + f(n)$$

The labels a, n, c, m, f can of course be "anything" but otherwise any other form will likely cause an error.

The method of undetermined coefficients which we will illustrate below tells us if

```
> f(n) = P(n)*b^n;
```

$$\mathrm{f}(n) = \mathrm{P}(n)\, b^n$$

where P(n) is a polynomial of degree t then there is a particular solution of the form

```
> a(n) = n^s*Q(n)*b^n;
```

$$\mathrm{a}(n) = n^s\, \mathrm{Q}(n)\, b^n$$

where Q(n) is a polynomial of degree t with coefficients to be determined, s = 0 if b is not a characteristic root and s is the multiplicity of b otherwise.

If we combine this method with the superposition principle we can solve quite a few simple linear recurrence equations.

The Associated Homogeneous Equation

First we need a procedure to compute the associated homogeneous recurrence equation, that is, to remove the inhomogeneous term.

```
> hom:=proc(eqn)
>   local inhom,a;
>   a:=op(0,lhs(eqn));
>   inhom:=simplify(subs(a=0,rhs(eqn)));
>   return(lhs(eqn)=simplify(rhs(eqn)-inhom));
> end:
```

Let's check it.

```
> hom(eqn0);
```

$$a(n) = 4\ a(n-1) - 4\ a(n-2)$$

The Characteristic Polynomial

Next we want to compute the characteristic polynomial of a homogeneous recurrence relation. We normalize it to be monic for aesthetic reasons (and also so it will be unique).

```
> cxpoly:=proc(eqn,z)
>   local p,q,n,k,a;
>   n:=op(1,lhs(eqn));
>   a:=unapply(lhs(eqn),n);
>   p:=subs(a=(k->z^k),lhs(eqn)-rhs(eqn));
>   p:=sort(simplify(p),z);
>   q:=op(p)[nops(p)];
>   p:=simplify(p/q);
> end:
> cpoly:=(eqn,z)->monic(cxpoly(eqn,z),z):
```

Let's check it.

```
> cpoly(hom(eqn0),z); roots(%);
```

$$z^2 - 4\ z + 4$$

$$[[2, 2]]$$

The return from roots() here means that 2 is a root of multiplicity 2. In particular 3 and 4 are not a characteristic roots and so we have a paricular solution eqn0 of the form a(n) = A 3^n + B 4^n. To compute A and B we substitue our trial solution in eqn0 and then solve for A and B.

Substitution of Trial Solution

```
> rsubs:=proc(sb,eqn)
>   local e,n;
>   n:=op(1,lhs(sb));
>   e:=subs(unapply(lhs(sb),n)=unapply(rhs(sb),n), eqn);
>   simplify(e);
> end:
```

Let's try it

```
> subexpr:=rsubs(a(n)=A*3^n+B*4^n, eqn0);
```

$$subexpr := A\,3^n + B\,4^n = \frac{8}{9} A\,3^n + 3\,B\,4^{(n-1)} + 3^n + 4^n$$

The result of the substitution is an identity in n. Thus we take special values of n to obtain a set of equations to solve for A and B.

```
> xeqn0:={subs(n=0,subexpr),subs(n=1,subexpr)};
```

$$xeqn0 := \{\,3\,A + 4\,B = \frac{8}{3} A + 3\,B + 7,\ A + B = \frac{8}{9} A + \frac{3}{4} B + 2\,\}$$

```
> solve(xeqn0,{A,B});
```

$$\{\,B = 4,\ A = 9\,\}$$

Example 1

```
> eqn1:=a(n)=4*a(n-1)-4*a(n-2)+n*2^n+n*5^n;
```

$$eqn1 := \mathrm{a}(n) = 4\,\mathrm{a}(n-1) - 4\,\mathrm{a}(n-2) + n\,2^n + n\,5^n$$

```
> eqn1h:=hom(eqn1);
```

$$eqn1h := \mathrm{a}(n) = 4\,\mathrm{a}(n-1) - 4\,\mathrm{a}(n-2)$$

```
> cp1:=cpoly(eqn1h,z); roots(cp1,z);
```

$$cp1 := z^2 - 4\,z + 4$$

$$[\,[\,2,\,2\,]\,]$$

Thus 2 is characteristic with multiplicity 2, and 5 is noncharacteristic. Hence we expect a particular solution of the form

```
> trial1:=a(n)=n^2*(A1+A2*n)*2^n + (A3+A4*n)*5^n;
```

$$trial1 := a(n) = n^2 (A1 + A2\, n)\, 2^n + (A3 + A4\, n)\, 5^n$$

```
> subex1:=rsubs(trial1,eqn1);
```

$$subex1 := n^2\, 2^n\, A1 + n^3\, 2^n\, A2 + 5^n\, A3 + 5^n\, A4\, n = n^2\, 2^n\, A1 + n^3\, 2^n\, A2 - 6\, 2^n\, A2\, n - 2^{(n+1)}\, A1 + 6\, 2^n\, A2 + \frac{16}{25}\, 5^n\, A3 + \frac{16}{25}\, 5^n\, A4\, n - \frac{12}{25}\, 5^n\, A4 + n\, 2^n + n\, 5^n$$

```
> subeq1:={seq(subs(n=k,subex1),k=0..3)};
```

$$subeq1 := \{A3 = -2\, A1 + 6\, A2 + \frac{16}{25}\, A3 - \frac{12}{25}\, A4,$$
$$2\, A1 + 2\, A2 + 5\, A3 + 5\, A4 = -2\, A1 + 2\, A2 + \frac{16}{5}\, A3 + \frac{4}{5}\, A4 + 7,$$
$$16\, A1 + 32\, A2 + 25\, A3 + 50\, A4 = 8\, A1 + 8\, A2 + 16\, A3 + 20\, A4 + 58,$$
$$72\, A1 + 216\, A2 + 125\, A3 + 375\, A4 = 56\, A1 + 120\, A2 + 80\, A3 + 180\, A4 + 399\}$$

```
> parm1:=solve(subeq1,{A1,A2,A3,A4});
```

$$parm1 := \{A2 = \frac{1}{6}, A3 = \frac{-100}{27}, A4 = \frac{25}{9}, A1 = \frac{1}{2}\}$$

```
> soln1:=rhs(subs(parm1,trial1));
```

$$soln1 := n^2 \left(\frac{1}{2} + \frac{1}{6}\, n\right) 2^n + \left(-\frac{100}{27} + \frac{25}{9}\, n\right) 5^n$$

Let's compare our solution with the direct Maple solution

```
> msoln1:=rsolve(eqn1,a);
```

$$msoln1 := \left(-a(0) + \frac{1}{2}\, a(1)\right)(n+1)\, 2^n - \left(\frac{1}{2}\, a(1) - 2\, a(0)\right) 2^n + (n+1)\left(\frac{1}{2}\, n + 1\right)\left(1 + \frac{1}{3}\, n\right) 2^n - (n+1)\left(\frac{1}{2}\, n + 1\right) 2^n + \left(-\frac{43}{18}\, n - \frac{43}{18}\right) 2^n + \frac{329}{54}\, 2^n + \left(\frac{25}{9}\, n + \frac{25}{9}\right) 5^n - \frac{175}{27}\, 5^n$$

```
> diff1:=simplify(soln1-msoln1);
```

$$diff1 := 2^n\, a(0)\, n - 2^n\, a(0) - 2^{(n-1)}\, a(1)\, n + \frac{37}{18}\, n\, 2^n - \frac{100}{27}\, 2^n$$

It remains to show all the terms here are solutions of the homogeneous equation.

```
> sub1:=rsubs(a(n)=diff1,eqn1h): lhs(sub1)-rhs(sub1);
```

$$0$$

Example 2

```
> eqn2:=a(n)=9*a(n-1)-9*a(n-2)-41*a(n-3)+42*a(n-4)+(2+5*n^3)*7^n
  -2+n*4^n;
```

$$eqn2 := a(n) = 9\,a(n-1) - 9\,a(n-2) - 41\,a(n-3) + 42\,a(n-4) + (2 + 5\,n^3)\,7^n - 2 + n\,4^n$$

```
> eqn2h:=hom(eqn2);
```

$$eqn2h := a(n) = 9\,a(n-1) - 9\,a(n-2) - 41\,a(n-3) + 42\,a(n-4)$$

```
> cp2:=cpoly(eqn2h,z);
```

$$cp2 := z^4 - 9\,z^3 + 9\,z^2 + 41\,z - 42$$

```
> roots2:=roots(cp2);
```

$$roots2 := [[1, 1], [3, 1], [7, 1], [-2, 1]]$$

We note 7 and 1 are characteristic roots of multiplicity 1, and 4 is not a characteristic root. Thus we expect a solution of the form

```
> trial2:=a(n)=n*(A1+A2*n+A3*n^2+A4*n^3)*7^n+n*A5+(A6+A7*n)*4^n;
```

$$trial2 := a(n) = n\,(A1 + A2\,n + A3\,n^2 + A4\,n^3)\,7^n + n\,A5 + (A6 + A7\,n)\,4^n$$

```
> subex2:=rsubs(trial2,eqn2);
```

$$subex2 := n\,7^n\,A1 + n^2\,7^n\,A2 + n^3\,7^n\,A3 + n^4\,7^n\,A4 + n\,A5 + 4^n\,A6 + 4^n\,A7\,n = -2 - 36\,A5 + 2\,7^n + \frac{9}{64}\,4^n\,A7 - \frac{216}{343}\,7^n\,A1 - \frac{12}{49}\,7^n\,A2 + \frac{786}{343}\,7^n\,A3 - 48\,7^{(n-1)}\,A4 + n\,A5 + \frac{155}{128}\,4^n\,A6 + n\,4^n + 5\,7^n\,n^3 + n\,7^n\,A1 - \frac{432}{343}\,7^n\,A2\,n - \frac{648}{343}\,7^n\,A3\,n^2 - \frac{36}{49}\,7^n\,A3\,n - \frac{864}{343}\,7^n\,A4\,n^3 - \frac{72}{49}\,7^n\,A4\,n^2 + \frac{3144}{343}\,7^n\,A4\,n + \frac{155}{128}\,4^n\,A7\,n + n^4\,7^n\,A4 + n^3\,7^n\,A3 + n^2\,7^n\,A2$$

```
> subeq2:={seq(subs(n=k,subex2),k=0..6)};
```

$$subeq2 := \{ 705894\,A1 + 4235364\,A2 + 25412184\,A3 + 152473104\,A4 + 6\,A5 + 4096\,A6 + 24576\,A7 = 127320792 - 30\,A5 + 30336\,A7 + 631806\,A1 + 3317496\,A2 + 17161662\,A3 + 87901296\,A4 + 4960\,A6,\ 84035\,A1 + 420175\,A2 + 2100875\,A3 + 10504375\,A4 + 5\,A5 + 1024\,A6 + 5120\,A7 = 10543107 - 31\,A5 + 6344\,A7 + 73451\,A1 + 310219\,A2 + 1283849\,A3 + 5250007\,A4 + 1240\,A6,\ 1029\,A1 + 3087\,A2 + 9261\,A3 + 27783\,A4 + 3\,A5 + 64\,A6 + 192\,A7 = 47181 - 33\,A5 + \frac{483}{2}\,A7 + 813\,A1 + 1707\,A2 + 3459\,A3 + 6999\,A4 + \frac{155}{2}\,A6,\ 9604\,A1 + 38416\,A2 + 153664\,A3 + 614656\,A4 + 4\,A5 + 256\,A6 + 1024\,A7 = 774144 - 32\,A5 + 1276\,A7 + 8092\,A1 + 25732\,A2 + 79534\,A3 + 242704\,A4 + 310\,A6,\ 98\,A1 + 196\,A2 + 392\,A3 + 784\,A4 + 2\,A5 + 16\,A6 + 32\,A7 =$$

$$2088 - 34\,A5 + 41\,A7 + \frac{470}{7}\,A1 + \frac{424}{7}\,A2 + 62\,A3 + \frac{496}{7}\,A4 + \frac{155}{8}\,A6,$$

$$7\,A1 + 7\,A2 + 7\,A3 + 7\,A4 + A5 + 4\,A6 + 4\,A7 =$$

$$51 - 35\,A5 + \frac{173}{32}\,A7 + \frac{127}{49}\,A1 - \frac{173}{49}\,A2 + \frac{229}{49}\,A3 - \frac{233}{49}\,A4 + \frac{155}{32}\,A6,$$

$$A6 = -36\,A5 + \frac{9}{64}\,A7 - \frac{216}{343}\,A1 - \frac{12}{49}\,A2 + \frac{786}{343}\,A3 - \frac{48}{7}\,A4 + \frac{155}{128}\,A6\}$$

```
> parm2:=solve(subeq2,{A1,A2,A3,A4,A5,A6,A7});
```

$$parm2 := \left\{A7 = \frac{-128}{27},\, A6 = \frac{256}{81},\, A4 = \frac{1715}{864},\, A3 = \frac{-12005}{7776},\, A2 = \frac{1432025}{93312},\, A5 = \frac{-1}{18},\, A1 = \frac{-50428889}{1679616}\right\}$$

```
> soln2:=rhs(subs(parm2,trial2));
```

$$soln2 := n\left(-\frac{50428889}{1679616} + \frac{1432025}{93312}\,n - \frac{12005}{7776}\,n^2 + \frac{1715}{864}\,n^3\right)7^n - \frac{1}{18}\,n + \left(\frac{256}{81} - \frac{128}{27}\,n\right)4^n$$

Recall this is just a particular solution. Here is Maple's complete solution for comparison

```
> msoln2:=rsolve(eqn2,a);
```

$$msoln2 := \frac{1}{36}\,\mathrm{a}(1) + \frac{7}{6}\,\mathrm{a}(0) - \frac{2}{9}\,\mathrm{a}(2) + \frac{1}{36}\,\mathrm{a}(3) - \left(\frac{7}{20}\,\mathrm{a}(0) - \frac{9}{40}\,\mathrm{a}(1) - \frac{3}{20}\,\mathrm{a}(2) + \frac{1}{40}\,\mathrm{a}(3)\right)3^n$$
$$+ \left(-\frac{31}{135}\,\mathrm{a}(1) + \frac{7}{45}\,\mathrm{a}(0) + \frac{11}{135}\,\mathrm{a}(2) - \frac{1}{135}\,\mathrm{a}(3)\right)(-2)^n$$
$$- \left(\frac{5}{216}\,\mathrm{a}(1) - \frac{1}{36}\,\mathrm{a}(0) + \frac{1}{108}\,\mathrm{a}(2) - \frac{1}{216}\,\mathrm{a}(3)\right)7^n - \frac{1}{18}\,n - \frac{10654513}{7776} + \frac{6672507}{5120}\,3^n$$
$$+ \frac{94479203}{295245}\,(-2)^n + \frac{1715}{36}\,(n+1)\left(\frac{1}{2}\,n + 1\right)\left(1 + \frac{1}{3}\,n\right)\left(\frac{1}{4}\,n + 1\right)7^n$$
$$- \frac{166355}{1296}\,(n+1)\left(\frac{1}{2}\,n + 1\right)\left(1 + \frac{1}{3}\,n\right)7^n + \frac{6926885}{46656}\,(n+1)\left(\frac{1}{2}\,n + 1\right)7^n$$
$$+ \left(-\frac{195919199}{1679616}\,n - \frac{195919199}{1679616}\right)7^n - \frac{12535158187}{60466176}\,7^n + \left(-\frac{128}{27}\,n - \frac{128}{27}\right)4^n + \frac{640}{81}\,4^n$$

Bernoulli Sums

The Bernoulli sum

```
> Sum(k^m,k=1..n);
```

$$\sum_{k=1}^{n} k^m$$

is the solution of the recurrence equation

```
> eqn3:=a(n)=a(n-1)+n^m; init3:=a(0)=0;
```

$$eqn3 := a(n) = a(n-1) + n^m$$

$$init3 := a(0) = 0$$

Since

```
> cp3:=cpoly(hom(eqn3),z);
```

$$cp3 := z - 1$$

we see 1 is characteristic with multiplicity 1. Thus we expect a particular solution of the form a(n) = n*P(n) where P(n) is a polynomial of degree m. The solution of the homogeneous equation is obviously just a constant. Thus the general solution is c + n*P(n). Substituting the initial condition yields c = 0. Thus

```
> Sum(k^m,k=1..n)=n*P(n);
```

$$\sum_{k=1}^{n} k^m = n\,P(n)$$

where P(n) is a polynomial of degree m. As above we determine the coefficients of P(n) from a *finite* number of cases.

For m = 4 we have

```
> trial3:=a(n)=n*(A1+A2*n+A3*n^2+A4*n^3+A5*n^4);
```

$$trial3 := a(n) = n\,(A1 + A2\,n + A3\,n^2 + A4\,n^3 + A5\,n^4)$$

```
> subex3:=rsubs(trial3,subs(m=4,eqn3));
```

$$subex3 := n\,(A1 + A2\,n + A3\,n^2 + A4\,n^3 + A5\,n^4) = A2 - A3 - A5 - 5\,A5\,n^4 - 2\,A2\,n - 4\,A4\,n + A4 + 3\,A3\,n + 6\,A4\,n^2 + 10\,A5\,n^3 - 10\,A5\,n^2 + n\,A1 + A2\,n^2 + A3\,n^3 + A4\,n^4 + A5\,n^5 - 3\,A3\,n^2 - 4\,A4\,n^3 + 5\,n\,A5 - A1 + n^4$$

```
> subeq3:={seq(subs(n=k,subex3),k=0..4)};
```

$$subeq3 := \{3\,A1 + 9\,A2 + 27\,A3 + 81\,A4 + 243\,A5 = 4\,A2 + 8\,A3 + 32\,A5 + 16\,A4 + 2\,A1 + 81,\ 2\,A1 + 4\,A2 + 8\,A3 + 16\,A4 + 32\,A5 = A2 + A3 + A5 + A4 + A1 + 16,\ 0 = A2 - A3 - A5 + A4 - A1\ A1 + A2 + A3 + A4 + A5 = 1,\ 4\,A1 + 16\,A2 + 64\,A3 + 256\,A4 + 1024\,A5 = 9\,A2 + 27\,A3 + 243\,A5 + 81\,A4 + 3\,A1 + 256\}$$

```
> parm3:=solve(subeq3,{A1,A2,A3,A4,A5});
```

$$parm3 := \{A2 = 0, A5 = \frac{1}{5}, A3 = \frac{1}{3}, A1 = \frac{-1}{30}, A4 = \frac{1}{2}\}$$

```
> soln3:=rhs(subs(parm3,trial3));
```

$$soln3 := n\left(-\frac{1}{30} + \frac{1}{3}n^2 + \frac{1}{2}n^3 + \frac{1}{5}n^4\right)$$

Of course, we can evaluate the Bernoulli sums directly in Maple

```
> bern4:=sum(k^4,k=1..n);
```

$$bern4 := \frac{1}{5}(n+1)^5 - \frac{1}{2}(n+1)^4 + \frac{1}{3}(n+1)^3 - \frac{1}{30}n - \frac{1}{30}$$

```
> simplify(soln3-bern4);
```

$$0$$

Problems

Use the method of undetermined coefficients as above to find particular solutions

Problem 1

```
> a(n)=4*a(n-1)+4*a(n-2)+1+n+2^n+n^2*3^n;
```

$$a(n) = 4\, a(n-1) + 4\, a(n-2) + 1 + n + 2^n + n^2\, 3^n$$

Problem 2

```
> a(n)=8*a(n-1)-17*a(n-2)-6*a(n-3)+44*a(n-4)-8*a(n-5)-32*a(n-6)+n*2^
  n+n*4^n+n*(-1)^n;
```

$$a(n) = 8\, a(n-1) - 17\, a(n-2) - 6\, a(n-3) + 44\, a(n-4) - 8\, a(n-5) - 32\, a(n-6) + n\, 2^n + n\, 4^n + n\,(-1)^n$$

Problem 3

```
> a(n)=a(n-4)+n;
```

$$a(n) = a(n-4) + n$$

Problem 4

```
> a(n)=a(n-2)+n^4*2^n;
```

$$a(n) = a(n-2) + n^4\, 2^n$$

Problem 5

```
> a(n)=-a(n-2)+n^4*2^n;
```

$$a(n) = -a(n-2) + n^4\, 2^n$$

Problem 6

```
> a(n)=a(n-2)+(n^4-n^2)*2^n;
```

$$a(n) = a(n-2) + (n^4 - n^2)\, 2^n$$

Problem 7

```
> a(n) = 9*a(n-1)-27*a(n-2)+27*a(n-3)+n^3*(3^n+4^n);
```

$$a(n) = 9\,a(n-1) - 27\,a(n-2) + 27\,a(n-3) + n^3\,(3^n + 4^n)$$

>